
Need for Data-Level Parallelism

• Many real-world applications are
compute-intensive and data-parallel

• Need to process a lot of data, mostly
floating-point operations

• Examples are
▶ Real-time high-definition graphics

applications such as your favorite
video games

▶ Iterative kernels which update
elements according to some fixed
pattern called a stencil

Rise of GPGPU Computing

• Researchers used to trick GPUs to perform non-rendering computations, often
referred to as general-purpose GPU (GPGPU) computations

• Programming initial GPU devices for other purposes was very convoluted
− Programming model was very restrictive
− Examples of limitations: input colors and texture units, writes to arbitrary locations, and

floating-point computations
• This spurred the need for a generic highly-parallel computational device with high

computational power and memory bandwidth
▶ CPUs are more complex devices catering to a wider audience

Rise of GPGPU Computing

• NVIDIA released GeForce 8800 GTX in 2006 with CUDA architecture
+ General-purpose ALU and instruction set for general-purpose computation
+ Allowed arbitrary reads and writes to memory
+ IEEE compliance for single-precision floating-point arithmetic

• Introduced CUDA C and the toolchain for ease of development with the CUDA
architecture

• GPUs are now popularly used across several application domains
▶ For example, game effects, computational science simulations, image processing,

machine learning, and linear algebra
▶ Modern GPUs serve both as a programmable graphics processor and a scalable parallel

computing platform

• There are several GPU vendors like NVIDIA, AMD, Intel, Qualcomm, and ARM

GPU Architecture

Philosophy and Design Goals



Analytical Model to Compare CPU and GPU Performance

Simple cache model where threads do not share data and there is infinite off-chip memory
bandwidth

Large cache shared
among few threads

Working set no longer
fits in the cache

Hides long off-chip
latency

multicore
CPUs

multithreading
with GPUs

Philosophy of GPUs

• GPUs are suited for compute-intensive data-parallel applications
▶ The same program is executed for each data element

• Many-core chip
▶ Multi-threaded execution on a single core (multiple threads executed concurrently by a

core)
▶ SIMD execution within a single core (many ALUs performing the same instruction)

• The focus is on overall computing throughput rather than on the speed of an
individual core
▶ GPUs do not reduce latency, they aim to hide latency
▶ High arithmetic intensity and large number of schedulable units aim to hide the latency

of memory accesses

Key Insights in GPU Architecture
Much more real-estate (i.e., transistors) is devoted to computation rather than data
caching and control flow

Comparing FLOPS

Memory Bandwidth for CPU and GPU

Compare GPU to CPU Architecture

• CPUs aim to reduce memory latency with increasingly large and complex memory
hierarchy

• Disadvantages
▶ The Intel I7-920 processor has some 8 MB of internal L3 cache, almost 30% of the size

of the chip
− Larger cache structures increases the physical size of the processor
− Implies more expensive manufacturing costs and increases likelihood of manufacturing

defects

• Effect of larger, progressively more inefficient caches ultimately results in higher costs
to the end user

Advantages of a GPU

Performance

• Comparing Xeon 8180M and Titan V
(based on peak values)
▶ 3.4X operations executed per second

compared to the CPU
▶ 5.5X bytes transferred from main

memory per second compared to the
CPU

▶ Cost- and energy-efficiency
■ 15X as much performance per dollar
■ 2.8X as much performance per watt

Energy

• GPU’s higher performance and energy
efficiency are due to different allocation
of chip area
▶ High degree of SIMD parallelism,

simple in-order cores, less
control/sync. logic, less
cache/scratchpad capacity

▶ SIMD is more energy-efficient than
MIMD since a single instruction can
launch many data operations

▶ Simpler pipeline with no support for
restartable instructions and precise
exceptions

Limitations of GPUs

Should we not use GPUs ALL the time?
• GPUs can only execute some types of code fast

▶ SIMD parallelism is not well suited for all algorithms
▶ Need lots of data parallelism, data reuse, and regularity

• GPUs are harder to program and tune than CPUs because of their architecture
▶ Fewer tools and libraries exist

Role of CPUs
• CPU is responsible for initiating computation on the GPU and transferring data to and

from the GPU
• Beginning and end of the computation typically require access to input/output (I/O)

devices
• There are ongoing efforts to develop APIs providing I/O services directly on the GPU

▶ GPUs are not standalone yet, assumes the existence of a CPU

CPUs vs GPUs

CPU

• Designed for running a few potentially
complex tasks
▶ Tasks may be unconnected
▶ Suitable to run system software like

the OS and applications
• Small number of registers per core

private to a task
▶ Context switch between tasks is

expensive in terms of time
▶ Register set must be saved to memory

and the next one restored from
memory

GPU

• Designed for running large number of
simple tasks
▶ Suitable for data parallelism

• Has a single set of registers but with
multiple banks
▶ A context switch involves setting a

bank selector to switch in and out the
current set of registers

▶ Orders of magnitude faster than
having to save to RAM

GPU Architecture
• GPUs consist of Streaming Multiprocessors (SMs)

▶ NVIDIA calls these streaming multiprocessors and AMD calls them compute units
• SMs contain Streaming Processors (SPs) or Processing Elements (PEs)

▶ Each core contains one or more ALUs and FPUs
• GPU can be thought of as a multi-multicore (manycore) system

Ampere Architecture

• A GA102 GPU includes 28.3 billion transistors with a die size of 628.4 mm2

• Includes 7 Graphics Processing Clusters (GPCs), 42 (7*6) Texture Processing Clusters
(TPCs), and 84 (7*12) Streaming Multiprocessors (SMs)

• Each SM in GA10x GPUs contain
▶ 128 CUDA cores for a total of 84*128=10752 cores
▶ 256 (4*16384*32 bits) KB register file
▶ Combined 128 KB L1 data cache/shared memory subsystem

• In addition, there are 84 RT Cores, 336 Tensor Cores, 168 FP64 units (two per SM)
• The memory subsystem consists of twelve 32-bit memory controllers (384-bit total)

▶ 512 KB of L2 cache is paired with each 32-bit memory controller, for a total of 6144 KB

• Includes PCIe Gen4 providing up to 16 Gigatransfers/second bit rate

NVIDIA Ampere GA102 GPU Architecture: Second-Generation RTX

Ampere Architecture

Ampere Architecture
Each SM contains 128 CUDA cores, 4 Tensor cores, 4
Texture units, 1 Ray Tracing core, a 256 KB register
file, and 128 KB of L1/Shared memory.

An SM is partitioned into four processing blocks
(sub-cores), each with a 64 KB register file, an L0
instruction cache, one warp scheduler, one dispatch
unit, and sets of math and other units.

L1 (KB) Shared memory (KB)
128 0
120 8
112 16
96 32
64 64
28 100

Swarnendu Biswas (IIT Kanpur) CS 610: GPU Architecture and CUDA Programming Sem 2026-27-I 18 / 152



More on Ampere Architecture

• Each SM processing block (also called partition or sub-core) has two datapaths
• One datapath in each partition consists of 16 FP32 CUDA Cores
• The second datapath consists of both 16 FP32 CUDA Cores and 16 INT32 Cores
• A GA102 SM can execute either 128 FP32 operations or 64 FP32 + 64 INT32

operations per clock

• Partitioning the SM allows for better scalability
• Each sub-core can issue instructions independently for its own pipeline, thereby

increasing instruction throughput
• Reduces contention on the register file and interconnects, can be independently

clock- or power-gated

Blackwell Architecture

• A GB202 GPU includes 92.2 billion transistors with a die size of 750 mm2

• Includes 12 Graphics Processing Clusters (GPCs), 96 (12*8) Texture Processing
Clusters (TPCs), 192 (12*16) Streaming Multiprocessors (SMs)

• Each SM in GB20x GPUs contain
▶ 128 CUDA cores for a total of 192*128=24576 cores
▶ 256 (4*16384*32 bits) KB register file
▶ Combined 128 KB L1 data cache/shared memory subsystem

• In addition, there are 192 RT Cores, 768 Tensor Cores, and 384 FP64 units (two per
SM)

• The memory subsystem consists of 16 32-bit memory controllers (512-bit total)
▶ 8192 KB of L2 cache is paired with each 32-bit memory controller, for a total of 128 MB

• Includes PCIe Gen5 providing up to 32 Gigatransfers/second bit rate

NVIDIA RTX Blackwell GPU Architecture: Built for Neural Rendering

Compute Capability (CC)

• When programming with CUDA, it is very important to be aware of the differences
among different versions of hardware

• In CUDA, CC refers to architecture features
▶ For example, number of registers and cores, cache and memory size, and supported

arithmetic instructions

• For example, CC 1.x devices have 16 KB local memory per thread, and 2.x and 3.x
devices have 512 KB local memory per thread

CUDA GPU Compute Capability
CUDA: Version features and specifications

Discrete vs Integrated GPUs

Discrete
• More performant, consumes more

energy
• Cost of PCIe transfers influences the

granularity of offloading and the
performance

CPU GPU

CPU
Memory

GPU
Memory

PCIe bus

Integrated
• Less performant because of energy

considerations
• CPU and GPU share physical memory

(DRAM or LLC) and can avoid the cost of
data transfers over a PCIe bus

CPU GPU

GPU
Memory

Cache

CUDA Programming

Programming NVIDIA GPUs



CUDA Philosophy

Massively parallel
The computations can be broken down into hundreds or thousands of independent units
of work

Computationally intensive
The time spent on computation significantly exceeds the time spent on transferring data to
and from GPU memory

Single Instruction Multiple Thread (SIMT)

CUDA Programming Model

Allows fine-grained data parallelism and
thread parallelism nested within
coarse-grained data parallelism and task
parallelism
• Independent grids (i.e., kernels) represent task

parallelism
• Concurrent thread blocks represent

coarse-grained data parallelism
• Concurrent threads/warps represent

fine-grained data parallelism or thread
parallelism

GPU
Application

Task 1
(kernel 1)

Task n
(kernel n)

.. .. ..

.. .. ..

.. .. ..

Thread block

G
rid

.. .. ..

.. .. ..

.. .. ..

Thread block

G
rid

(i) Partition the problem into coarse sub-problems that can be solved independently
(ii) Assign each sub-problem to a “block” of threads to be solved in parallel

(iii) Each sub-problem is also decomposed into finer work items that are solved in parallel by all
threads within the block

Swarnendu Biswas (IIT Kanpur) CS 610: GPU Architecture and CUDA Programming Sem 2026-27-I 24 / 152



Hello World with CUDA

#include <stdio.h>
#include <cuda.h>

__global__ void hwkernel() {
printf("Hello world!\n");

}

int main() {
hwkernel<<<1, 1>>>();

}

$ nvcc hello-world.cu
$ ./a.out

$

hello-world.cu Makefile

Hello World with CUDA

#include <stdio.h>
#include <cuda.h>

__global__ void hwkernel() {
printf("Hello world!\n");

}

int main() {
hwkernel<<<1, 1>>>();
cudaDeviceSynchronize();

}

$ nvcc hello-world.cu
$ ./a.out
Hello world!

$

• CPU thread returns immediately after launching the kernel
• Use cudaDeviceSynchronize() (or its variants) to block the caller CPU thread

Function Declarations

Executed on Callable from

__device__ float deviceFunc() Device Device
__global__ void kernelFunc() Device Host§

__host__ float hostFunc() Host Host

• __global__ defines a kernel function, must return void
• __device__ functions can have return values
• __host__ is default, and can be omitted
• Prepending __host__ __device__ causes the system to compile separate host and device

versions of the function

§A kernel function can also be called from the device if dynamic parallelism is enabled

Classical Copy-Then-Execute Model

(i) Prepare input data in CPU memory
(ii) Copy data from CPU to GPU memory (e.g., cudaMemcpy())

(iii) Launch GPU kernel and execute
(iv) Copy results from GPU memory to CPU memory (e.g., cudaMemcpy())
(v) Use the results on the CPU

(vi) Repeat the above steps based on the application logic

Kernels

• Kernels are special functions that a CPU can call to execute on the GPU
▶ Executed N times in parallel by N different CUDA threads
▶ Cannot return a value

• Each thread has a unique thread ID that is accessible within the kernel through the
built-in threadIdx variable

__global__ void VecAdd(float* A, float* B, float* C) {
int i = threadIdx.x;
...

}
int main() {

auto *h_A = new float[N];
...
// Kernel invocation with 1 thread block and N threads per block
VecAdd<<<1, N>>>(d_A, d_B, d_C);

}

vector-addition.cu

Kernels

KernelName<<<m, n>>>(arg1, arg2, ...)

GPU spawns m blocks with n threads that run a copy of the same function
• CPU can continue processing while GPU runs the kernel
• Kernel call returns when all threads have terminated

kernel1<<<X,Y>>>(...);
// kernel starts execution, CPU continues to next statement
kernel2<<<X,Y>>>(...);
// kernel2 placed in queue, will start after kernel1 finishes,
// CPU continues
cudaMemcpy(...);
// CPU blocks until memory is copied, memory copy starts after all
// preceding CUDA calls finish

Thread Hierarchy
• A kernel executes in parallel across a set of parallel threads

▶ Threads that are generated by a kernel launch are collectively called a grid
• Threads are organized in thread blocks (aka cooperative thread arrays)

▶ A thread block is a set of concurrently executing threads that can communicate among
themselves through shared memory and barrier synchronization

• A grid is an array of thread blocks that execute the same kernel

Thread Hierarchy

• Threads in a block reside on the same
SM, max 1024 threads in a block

• Thread blocks are organized into 1D,
2D, or 3D grids
▶ Grid dimension is given by the

gridDim variable

• Identify block within a grid with the
blockIdx variable

• Block dimension is given by the
blockDim variable

Dimension and Index Variables

Dimension

• gridDim specifies the number of
blocks in the grid

• blockDim specifies the number of
threads in each block

Index

• blockIdx gives the index of the block
in the grid

• threadIdx gives the index of the
thread within the block

• The keywords are 3-component vectors
▶ For example, threadIdx.[x,y,z] gives the index of the thread in a block, in the

particular direction

Type is dim3

Finding Thread IDs
How to find out the linearized thread IDs within a block?
• 1D block: thread ID is threadIdx.x
• 2D block of size (Dx, Dy): thread ID of a thread of index (x, y) is (x + yDx)
• 3D block of size (Dx, Dy, Dz): thread ID of a thread of index (x, y, z) is (x +

yDx + zDxDy)

0 1 254 255... 0 1 254 255... 0 1 254 255...
Block 0 Block 1 Block 255

tid = blockIdx.x*blockDim.x
+ threadIdx.x 

tid = blockIdx.x*blockDim.x
+ threadIdx.x 

tid = blockIdx.x*blockDim.x
+ threadIdx.x 

...

...... ...

Variable tid is local
to each thread

thread-id.cu

Thread Block Clusters
CC 9.0 introduced an optional level in the thread hierarchy called Thread Block Clusters
• Thread blocks in a cluster are always co-scheduled on a GPC

▶ Thread blocks in a cluster can access the distributed Shared Memory
▶ Allows thread blocks in the cluster to synchronize with hardware support

• The number of thread blocks in a cluster can be specified with the compile-time
kernel attribute __cluster_dims__(X,Y,Z)

• A maximum of 8 thread blocks is guaranteed to be portable

Swarnendu Biswas (IIT Kanpur) CS 610: GPU Architecture and CUDA Programming Sem 2026-27-I 35 / 152



Execution Configuration Uses Integer Arithmetic

• Dimension variables are vectors of integral type
• Assume data is of length N, and say the kernel execution configuration is <<<N/TPB,

TPB>>>
▶ Each block has TPB threads and there are N/TPB blocks

Suppose N = 64 and TPB = 32

Implies there are 2 blocks of 32 threads

Suppose N = 65 and TPB = 32

Implies there are 2 blocks of 32 threads

• Ensure that the grid covers the array length by rounding up the number of blocks from
N/TPB to (N+TPB-1)/TPB

• Use a control statement in the kernel to ensure that the thread index is within the
maximum array index

Launching Kernels for 2D Data: Matrix Addition

// Kernel definition
__global__ void MatAdd(float A[N][N], float B[N][N], float C[N][N]) {
int j = blockIdx.x * blockDim.x + threadIdx.x;
int i = blockIdx.y * blockDim.y + threadIdx.y;
C[i][j] = A[i][j] + B[i][j];

}
int main() {
...
// Kernel invocation
dim3 threadsPerBlock(16, 16);
dim3 numBlocks(N / threadsPerBlock.x, N / threadsPerBlock.y);
MatAdd<<<numBlocks, threadsPerBlock>>>(A, B, C);
...

}

matadd.cu

Timing a CUDA Kernel

...
cudaEvent_t start = nullptr, stop = nullptr;
cudaEventCreate(&start);
cudaEventCreate(&stop);

cudaEventRecord(start,0);
MatMul<<<blocksPerGrid, threadsPerBlock>>>(d_A, d_B, d_C);
cudaEventRecord(stop);
cudaEventSynchronize(stop);

float kernel_time = 0.0F;
cudaEventElapsedTime(&kernel_time, start, stop);

cudaEventDestroy(start);
cudaEventDestroy(stop);
...

How to Implement Performance Metrics in CUDA C/C++

Choosing Optimal Execution Configuration

• The number of thread blocks in a grid is usually dictated by the size of the data being
processed or the number of processors in the system
▶ It is okay to have a greater number of threads

• Choose number of threads in a block to be a multiple of 32
• No fixed rule, needs exploration and experimentation to find the best grid

configuration
__global__ void MatMul(float* A, float* B, float* C) {

...
}
int main() {

...
constexpr uint32_t TB_DIM = 64;
dim3 threadsPerBlock(TB_DIM, TB_DIM);
dim3 blocksPerGrid((N + TB_DIM - 1) / TB_DIM, (N + TB_DIM - 1) / TB_DIM);
MatMul<<<blocksPerGrid, threadsPerBlock>>>(d_A, d_B, d_C);
...

}

matmul-naive-buggy.cu

Reporting Errors

• All CUDA API calls return an error code of type cudaError_t
▶ Error in the API call itself or error in an earlier asynchronous operation (e.g. kernel)

• Get the error code for the last error with cudaGetLastError()
• Get a string to describe the error with char
*cudaGetErrorString(cudaError_t)

What is the canonical way to check for errors using the CUDA runtime API?

Device Management

Query device information

• The application can query and select GPUs
▶ cudaGetDeviceCount(int *count)
▶ cudaSetDevice(int device)
▶ cudaGetDevice(int *device)
▶ cudaGetDeviceProperties(cudaDeviceProp *prop, int device)

• Multiple host threads can share a device
• A single host thread can manage multiple devices

▶ cudaSetDevice(i) to select current device
▶ cudaMemcpy(...) for peer-to-peer copies

Warp Scheduling



Understanding SIMT

• In vectorization, users program the SIMD hardware directly with vector ISA, or use
auto-vectorization or intrinsics
▶ We program with the vector width in mind, excepting auto-vectorization

• GPUs employ SIMD hardware to exploit data-level parallelism
• SIMT can be thought of as “SIMD with multithreading”

▶ Software analog compared to the hardware perspective of SIMD
▶ Distinct threads executing together can access non-contiguous memory or diverge into

different execution branches
• Multiple types of parallelism

▶ Independent grids (i.e., kernels) or concurrent thread blocks represent coarse-grained
data parallelism or task parallelism

▶ Concurrent threads/warps represent fine-grained data parallelism or thread parallelism

SM Microarchitecture

T. Aamodt et al. General-Purpose Graphics Processor Architectures. Springer Cham.

SM Microarchitecture Modeled in Accel-Sim

M. Khairy et al. Accel-Sim: An Extensible Simulation Framework for Validated GPU Modeling. ISCA 2020.
R. Huerta et a. Dissecting and Modeling the Architecture of Modern GPU Cores. MICRO 2025.

Mapping Blocks and Threads

• A GPU executes one or more kernel grids
• A kernel is partitioned into thread blocks that execute independently of each other
• When a CUDA kernel is launched, the thread blocks are distributed across the SMs

▶ Multiple thread blocks, up to a limit, can execute concurrently on one SM
■ Not all blocks may be resident at the same time

▶ For example, a CUDA device may allow up to eight blocks to be assigned to each SM
▶ CUDA cores in the SM execute threads of a block

• A block begins execution only when it has secured all execution resources necessary
for all the threads

• As thread blocks terminate, new blocks are launched on the vacated SMs

CUDA runtime can execute blocks in any order

• Blocks are mostly not supposed to synchronize with each other
▶ Allows for simple hardware support for data parallelism

Block Scalability

• A kernel with enough blocks scales
across GPUs

• A GPU with more SMs will automatically
execute the program faster than a GPU
with fewer SMs

Comparing the performance of vector addition kernel

GPU CC # SMs Time (ms)

GeForce GTX 1080 6.1 20 72.5
Quadro RTX 5000 7.5 48 8.6
A40 8.6 84 5.6

vector-addition.cu

Thread Warps

• Conceptually, threads in a block can also execute in any order
• However, sharing a control unit reduces hardware complexity, cost, and power

consumption
• A set (currently 32) of consecutive threads that execute in SIMD fashion is called a

warp
▶ Called wavefront (with 64 threads) on AMD GPUs

• Warps are scheduling units in an SM
▶ Implementation detail, not part of the programming model

• Warps issue their instructions in program order

Scheduling Thread Warps

• Each SM launches warps of threads, and executes warps on a time-sharing basis
• SM schedules and executes warps that are ready to run

▶ Warps run for fixed-length time slices like processes
▶ Warps whose next instruction has its operands ready for consumption are eligible for

execution
▶ If more than one warp is ready for execution, a greedy-then-youngest* priority

mechanism is used to select one for execution
■ Suppose an instruction to be executed by a warp has to wait for the result of a previously

initiated long-latency operation (e.g., ld.global)
■ The warp is not selected for execution, another warp that is not waiting for results is selected

for execution
▶ Selection of ready warps for execution does not introduce any idle time into the

execution timeline (i.e., zero-overhead scheduling)

*R. Huerta et a. Dissecting and Modeling the Architecture of Modern GPU Cores. MICRO 2025.

Scheduling Thread Warps

• Goal is to have enough warps so that the GPU scheduler is likely to always find a warp
ready to execute
▶ Hides latency of long memory operations with work from other threads (called latency

tolerance or hiding)
• Thread blocks execute on an SM, thread instructions execute on a core

▶ CUDA virtualizes the physical hardware
▶ Thread is a virtualized scalar processor (registers, PC, state)
▶ Block is a virtualized multiprocessor (threads, shared memory)

• As warps and thread blocks complete, resources are freed

Warp Scheduling
• For an ideal memory system, increasing the number of warps can increase the

throughput
▶ If the number of warps in a core multiplied by the issue time of each warp exceeds the

memory latency, the execution units in the core will always remain busy
▶ Round-robin scheduling of warps would suffice
▶ However, locality property either favors or discourages round-robin scheduling

• Increasing the number of warps is impractical because of the need to maintain
execution state in hardware (i.e., to achieve zero-overhead scheduling)
▶ Number of threads that can be simultaneously tracked and scheduled in hardware is

bounded
▶ Requires resources for an SM to maintain execution status of threads

• Up to 2048 threads can be assigned to each SM on recent CUDA devices
▶ For example, 8 blocks of 256 threads, or 4 blocks of 512 threads

• Assume a CUDA device with 28 SMs, where each SM can accommodate up to 2048
threads
▶ The device can have up to 57344 threads simultaneously residing in the device for

execution

Memory Hierarchy



Memory Hierarchy in NVIDIA A100

CUDA Refresher: The CUDA Programming Model

Memory Organization

• Host and device maintain their own
separate memory spaces
▶ A variable in CPU memory may

not be accessed directly in a GPU
kernel

• It is the programmer’s
responsibility to keep copies of
variables in sync

Memory Hierarchy in CUDA

Variable Type Qualifiers in CUDA

Memory Scope Lifetime

int localVar Register Thread Kernel
__device__ __local__ int localVar Local Thread Kernel
__device__ __shared__ int sharedVar Shared Block Kernel
__device__ int globalVar Global Grid Application
__device__ __constant__ int constVar Constant Grid Application

• __device__ is optional when used with __local__, __shared__, or __constant__

• CPU pointers can only point to memory allocated or declared in global memory

Registers

• 64K 32-bit registers (or 256 KB register file) per SM
▶ A CPU in contrast has a few (1–2 KB) per core

• Registers are allocated per warp, in chunks of 256
• Up to 255 registers per thread (CC 3.5+)
• If a code uses the maximum number of registers per thread (255) and an SM has 64K

registers, then the SM can support a maximum of 256 threads
• If we use the maximum allowable number of threads per SM (2048), then each thread

can use at most 32 registers per thread
• CUDA provides kernel qualifiers for passing additional information to nvcc for more

efficient register usage (e.g., __launch_bounds__() and __maxnreg__())

Registers

• 64K 32-bit registers (or 256 KB register file) per SM
▶ A CPU in contrast has a few (1–2 KB) per core

• Registers are allocated per warp, in chunks of 256
• Up to 255 registers per thread (CC 3.5+)
• If a code uses the maximum number of registers per thread (255) and an SM has 64K

registers, then the SM can support a maximum of 256 threads
• If we use the maximum allowable number of threads per SM (2048), then each thread

can use at most 32 registers per thread
• CUDA provides kernel qualifiers for passing additional information to nvcc for more

efficient register usage (e.g., __launch_bounds__() and __maxnreg__())

What if each thread uses 33 registers?

Shared Memory

• Each SM contains a shared memory structure that acts as a software-managed cache
• Shared memory (also called scratchpad memory) is used for efficient communication

among threads in a block
▶ Usually 16–100 KB of storage that can be accessed efficiently by all threads in a block
▶ The space is shared among all blocks running on that SM

Variable in shared memory is allocated
using the __shared__ specifier
• Latency is comparable to accessing

registers

Say an SM with 4 thread blocks has 16 KB of shared
memory

__shared__ float min[256];
__shared__ float max[256];
__shared__ float avg[256];
__shared__ float stdev[256];

Several things like grid configuration, number of registers, and amount of shared memory
per block limits occupancy

Using Shared Memory in CUDA C/C++

Dynamic Shared Memory
Kernels using shared memory over 48 KB per block must use dynamic shared memory
• Allocated at run time, and the size is specified as the third argument to the kernel

▶ Implies only one extern __shared__ array can be declared inside a kernel

__global__ void kernel(uint32_t dbl_arr_sz, uint32_t ch_arr_sz) {
extern __shared__ char array[];
double *dbl_array = (double *)array;
char *ch_array = array + dbl_arr_sz * sizeof(double);
...

}
int main() {

...
uint32_t DBL_ARR_SZ = 256, CH_ARR_SZ = 128;
uint32_t SHD_MEM_SZ = DBL_ARR_SZ * sizeof(double) + CH_ARR_SZ * sizeof(char);
kernel<<<1,1, SHD_MEM_SZ>>>(DBL_ARR_SZ, CH_ARR_SZ);
...

}

shmem-allocations.cu vector-reverse-shmem.cu

Global Memory

• Variable lock can be accessed by both
kernels
▶ Resides in global memory space
▶ Can be both read and modified by all

threads

__device__ int lock=0;
__global__ void kernel1(...) {

// Kernel code
}
__global__ void kernel2(...) {

// Kernel code
}

• On-device memory is accessed via 32, 64, or 128 B transactions
• A warp executes an instruction that accesses global memory

▶ The addresses are coalesced into transactions
▶ Number of transactions depend on the access size and distribution of memory addresses
▶ More transactions mean less throughput

■ For example, if 32 B transaction is needed for a thread’s 4 B access, throughput is essentially
1/8th

Memory Access Efficiency

Compute-to-global-memory access ratio is the number of floating-point operations
performed for each access to global memory

// A and B are pointers to float type
for (uint32_t k = 0; k < N; k++)
tmp += A[row*N+K]*B[k*N+col]);

Assume a GPU device with 1 TB/s global memory bandwidth and peak single-precision
performance of 12 TFLOPS
• What is the performance we expect with an access ratio of 1?
• We can do 1000/(4+4) GFLOPS, which is only ∼ 1% of the peak performance

Improving Matrix Multiplication with Tiling in Shared Memory

Sanket Shah. Hidden Speed in CUDA’s Shared Memory

Improving Matrix Multiplication with Tiling in Shared Memory
Phase 1

Thread0,0
Msh[0][0]
= M[0][0]

Nsh[0][0]
= N[0][0]

tmp0,0 +=
(Msh[0][0]*Nsh[0][0] + Msh[0][1]*Nsh[1][0])

Thread0,1
Msh[0][1]
= M[0][1]

Nsh[0][1]
= N[0][1]

tmp0,1 +=
(Msh[0][0]*Nsh[0][1] + Msh[0][1]*Nsh[1][1])

Thread1,0
Msh[1][0]
= M[1][0]

Nsh[1][0]
= N[1][0]

tmp1,0 +=
(Msh[1][0]*Nsh[0][0] + Msh[1][1]*Nsh[1][0])

Thread1,1
Msh[1][1]
= M[1][1]

Nsh[1][1]
= N[1][1]

tmp1,1 +=
(Msh[1][0]*Nsh[0][1] + Msh[1][1]*Nsh[1][1])

Phase 2

Thread0,0
Msh[0][0]
= M[0][2]

Nsh[0][0]
= N[2][0]

tmp0,0 +=
(Msh[0][0]*Nsh[0][0] + Msh[0][1]*Nsh[1][0])

Thread0,1
Msh[0][1]
= M[0][3]

Nsh[0][1]
= N[2][1]

tmp0,1 +=
(Msh[0][0]*Nsh[0][1] + Msh[0][1]*Nsh[1][1])

Thread1,0
Msh[1][0]
= M[1][2]

Nsh[1][0]
= N[3][0]

tmp1,0 +=
(Msh[1][0]*Nsh[0][0] + Msh[1][1]*Nsh[1][0])

Thread1,1
Msh[1][1]
= M[1][3]

Nsh[1][1]
= N[3][1]

tmp1,1 +=
(Msh[1][0]*Nsh[0][1] + Msh[1][1]*Nsh[1][1])

matmul-tiled-shmem.cu

Swarnendu Biswas (IIT Kanpur) CS 610: GPU Architecture and CUDA Programming Sem 2026-27-I 61 / 152

https://www.cse.iitk.ac.in/users/cs623/2026Autumn/lec623/part2/examples/cuda-intro/matmul-tiled-shmem.cu


Constant Memory

• Used for data that will not change during kernel execution
▶ Accessible from all threads within a grid
▶ Constant memory is 64 KB

// Global scope
__constant__ float d_filter[FILTER_WIDTH];
// Initialize array in constant memory
cudaMemcpyToSymbol(d_filter, h_filter, FILTER_WIDTH * sizeof(float));

• Constant memory is cached
▶ Each SM has a read-only constant cache that is shared by all cores in the SM
▶ Speeds up reads from the constant memory space which resides in GPU global memory
▶ Read from constant memory will incurs a global memory latency on the first miss
▶ Subsequent reads will hit in the constant cache, which is almost as fast as registers

Local Memory

• Local memory is off-chip per-thread memory
▶ More like thread-local global memory, so it requires memory transactions and consumes

bandwidth
• Automatic variables are placed in local memory

(i) Arrays when it is not known whether indices are constant quantities
(ii) Large structures or arrays that consume too much register space

(iii) In case of register spilling
• Inspect PTX assembly code (compile with –ptx)

▶ Check for ld.local and st.local mnemonic

Device Memory Management

• Global device memory can be allocated with cudaMalloc()
• Freed by cudaFree()
• Data transfer between host and device is with cudaMemcpy()
• Initialize memory with cudaMemset()

• There are asynchronous versions (e.g., cudaMemcpyAsync())

• Check the CUDA Programming Guide§ for additional APIs

§CUDA C++ Programming Guide: Device Memory

GPU Caches

• GPUs have private L1 and shared L2 data caches
▶ Texture and constant cache are available on all devices

• L1 and L2 cache lines are 128 B wide
• L1 caches in GPUs are not coherent, L2 cache is coherent

• Shared memory is partitioned out of unified data cache and its size can be configured,
remaining portion is the L1 cache

• On a global store, the L1 cache uses a write-through policy

• L2 cache is shared by all SMs
• Uses a writeback policy for global memory stores

More on GPU Caches

• GPU caches are sectored, where each sector is 32 B
• There is one tag for the entire cache line, but each sector has its own status bits (e.g.,

valid and dirty)
• If only one 32 B sector is modified, only that sector is written back
• On a cache miss, not all 4 sectors in the cache line have to be filled, just the ones that

need updating

+ Reduces tag array overhead and saves bandwidth in case of scattered accesses

Swarnendu Biswas (IIT Kanpur) CS 610: GPU Architecture and CUDA Programming Sem 2026-27-I 66 / 152



Cache Operators on Memory Instructions

• Accesses to global memory are cached both in L1 and L2 caches by default
• Use compile time options for finer control on the caching behavior
• Load from global memory

-dlcm=ca loads are cached in both L1 and L2 (default), default eviction policy
-dlcm=cg loads bypass L1 and are only cached in L2
-dlcm=cs hints that data accessed is streaming and marks the line for early eviction
-dlcm=cv refetches data from global memory on every load, bypassing L1 and

invalidating L2
-dlcm=lu hints that this is the final load for this address, marking it for early

eviction and avoids needless write-backs
• Store to global memory

-dscm=wb bypass or invalidate L1, L2 uses writeback, default eviction policy
-dscm=cg cache only in L2, bypassing L1
-dscm=cs marks the line for early eviction
-dscm=wt writes through the L2 cache to global memory

Configuring Shared Memory Size

• The portion of shared memory in the unified data cache is configurable per kernel
• The driver automatically configures the desired shared memory capacity for each

kernel to avoid occupancy bottlenecks
▶ For Volta, the unified data cache is of 128 KB, and the shared memory capacity can be

set to 0, 8, 16, 32, 64 or 96 KB
▶ For Turing, the unified data cache has a size of 96 KB, and the shared memory capacity

can be set to either 32 KB or 64 KB

• An user can provide hints with cudaFuncSetAttribute()

shared-mem-config.cu

CUDA Compilation

Binary compatibility of GPU applications is not guaranteed across different generations



How NVCC works

• Nvcc is a driver program based on LLVM

▶ Compiles and links all input files
▶ Requires a general-purpose C/C++

host compiler
▶ Uses GCC and G++ by default on Linux

platforms

NVIDIA CUDA Compiler Driver NVCC

Selected NVCC Options and File Types
Options Description
-std {c++17|. . . } Select a particular C++ dialect
-G Generate debug information for device code, turns off device opti-

mizations
-arch ARCH Specify the virtual GPU architecture for generating assembly
-code CODE Specify the real GPU architecture to assemble and optimize for

File Type Description
.cu CUDA source file
.c,.cpp,.cxx,.cc C/C++ source files
.ptx PTX intermediate assembly
.cubin CUDA device binary code for a single GPU architecture
.fatbin CUDA fat binary file that may contain multiple PTX and CUBIN files

CUDA Compilation Trajectory

(i) Input program is preprocessed for device compilation
(ii) It is compiled to a CUDA binary (.cubin) and/or PTX (Parallel Thread Execution)

intermediate code which are encoded in a fatbinary
(iii) Input program is processed for compilation of the host code
(iv) Synthesized host code and the embedded fatbinary are linked together to generate

the executable

• A compiled CUDA device binary includes
▶ Program text (instructions)
▶ Information about the resources required

■ N threads per block
■ X bytes of local data per thread
■ M bytes of shared space per block

CUDA Compilation Trajectory



Binary Application Compatibility

• NVIDIA does not guarantee binary compatibility of GPU applications across different
generations
▶ For example, a CUDA application compiled for a Fermi GPU will very likely not run on a

Turing GPU (and vice versa)
▶ Instruction set and instruction encodings of a generation is different from those of other

generations
• nvcc relies on a two stage compilation model for ensuring application compatibility

across GPU generations
▶ Code is compiled to a virtual assembly called PTX
▶ PTX code is assembled for a real GPU architecture

• Recommendation
▶ Specify a lower virtual architecture to improve portability and a higher real architecture

for improved performance

Two-Staged Compilation with Virtual and Real Architectures

PTX Assembly and SASS

• GPU compilation is performed via an intermediate representation PTX
▶ PTX is an acronym for Parallel Thread eXecution
▶ Goal is to provide an architecture-indent ISA for compilers
▶ Can be considered as assembly for a virtual GPU architecture

• Example PTX code

• SASS is the low-level assembly that compiles to binary microcode which executes
natively on NVIDIA GPUs
▶ SASS is an acronym for Source and ASSembly

Parallel Thread Execution ISA
A Gentle Introduction to CUDA PTX

Improving Application Portability with JIT Compilation

• Two-stage compilation does not help
improve application portability
▶ Generated code is bound to one GPU

generation (e.g., sm_53)
• There are two strategies to improve

portability
(i) nvcc will postpone assembly of PTX

code until application run time if only
a virtual GPU architecture is specified

(ii) Generate multiple translations for
different architectures and embed the
CUDA binaries in a fat binary

NVCC Examples

nvcc –arch=compute_30 –code=sm_52 hello-world.cu
Generate PTX assuming CC 3.0 and generate binary SASS code compliant with CC 5.2

nvcc –arch=compute_30 –code=sm_30,sm_52 hello-world.cu
Generate binary SASS for two GPU architectures and embed the cubin files in the executable

nvcc -arch=compute_50 hello-world.cu
Generate only PTX for CC 5.0 and embed in the binary

nvcc -arch=sm_52 hello-world.cu
• Implies nvcc -arch=compute_52 -code=sm_52,compute_52 hello-world.cu

• Embed both the PTX and the SASS code in the final binary

Synchronization in CUDA



Race Conditions and Data Races

• A race condition occurs when program behavior depends upon relative timing of two
(or more) event sequences

(i) Read value at address c
(ii) Add sum to value

(iii) Write result to address c
• There can be intra-warp, inter-warp, and inter-block races

▶ Intrawarp races occur when threads from the same warp write to the same memory
location

__global__ void intrawarp ( unsigned * data ) {
data [0] = threadIdx .x;

}

__syncthreads()

• __syncthreads() synchronizes threads within a block
• All global and shared memory accesses made by the participating threads prior to

__syncthreads() are visible to all threads in the block

• A __syncthreads() statement must be executed by all threads in a block
• If __syncthreads() is in an if statement, then either all threads in the block

execute the then path that includes the __syncthreads() or none of them does
• If __syncthreads() statement is in each path of an if-then-else statement,

then either all threads in a block execute the __syncthreads() on the then path or
all of them execute the else path
▶ The two __syncthreads() are different barrier synchronization points

syncthreads-deadlock1.cu
syncthreads-deadlock2.cu

Grid-level Synchronization

• cudaDeviceSynchronize()
synchronizes all threads in a grid
▶ There are other variants

• For threads from different grids, writes
from a kernel happen before reads
from subsequent grid launches

Atomic Operations

• Atomic operations allow performing atomic read-modify-write (RMW) operations on
data residing in global or shared memory
▶ Prevent data races associated with multiple threads concurrently accessing a global or

shared memory variable
▶ Will give predictable results when simultaneous access to memory is required
▶ For example, atomicAdd(), atomicSub(), atomicMin(), atomicMax(),

atomicInc(), atomicDec(), atomicExch(), atomicCAS()

Scoped Synchronization

• CUDA provides scope qualifiers that limit the subset of the threads that are
guaranteed to observe the synchronization

• CUDA exposes three scopes: block, device, and system
• For example, an atomic RMW operation with block scope is only visible to threads

within the same thread block (e.g., atomicAdd_block())

CUDA Memory Model

• A memory consistency model is a set of rules that govern how systems process
memory operation requests from multiple processors
▶ Determines the order in which memory operations appear to execute
▶ Specifies the allowed behaviors of multithreaded programs executing with shared

memory
▶ Memory models are defined both for hardware and programming languages

• CUDA implements a weakly-ordered memory model
• The order in which a thread writes data is not necessarily the order in which the data

is observed being written by another CUDA or host thread
• The behavior on concurrent and conflicting accesses is undefined

Swarnendu Biswas (IIT Kanpur) CS 610: GPU Architecture and CUDA Programming Sem 2026-27-I 83 / 152



Weakly-Ordered Memory Model in CUDA

__device__ int X = 1, Y = 2;

Thread 1

__device__ void writeXY() {
X = 10;
Y = 20;

}

Thread 2

__device__ void readXY() {
int B = Y;
int A = X;

}

A racy program has undefined behavior, and has no defined semantics

What are the possible
values of A and B?

Memory Fences

Semantics of __threadfence_block()
• All writes made by the calling thread before the call are observed by all threads in the block as

occurring before all writes made by the calling thread after the call
• All reads from all memory made by the calling thread before the call are ordered before all

reads from all memory made by the calling thread after the call

__device__ int X = 1, Y = 2;

Thread 1

__device__ void writeXY() {
X = 10;
__threadfence();
Y = 20;

}

Thread 2

__device__ void readXY() {
int B = Y;
__threadfence();
int A = X;

} What are the possible
values of A and B?

What can we say about the usage of __threadfence()?

__device__ int X = 1, Y = 2;

Thread 1

__device__ void writeXY() {
X = 10;
__threadfence();
Y = 20;

}

Thread 2

__device__ void readXY() {
int B = Y;
__threadfence();
int A = X;

}

Is using __threadfence() (i) always
correct and (ii) always required?

__syncthreads() vs __threadfence()

__syncthreads()

• Establishes visibility of memory
operations across threads

• Includes a barrier plus memory fence
functionality (i.e.,
__threadfence_block())

__threadfence()

• Ensures ordering of memory
operations by a thread

• __threadfence_block() does not
imply __syncthreads()

Warp-Level Primitives

• _ballot_sync(mask, predicate)
▶ Evaluate predicate for all non-exited

threads in mask and return an integer
whose Nth bit is set if predicate
evaluates to non-zero for the Nth
thread

• __activemask()
▶ Returns a 32-bit integer mask of all

currently active threads in the calling
warp Copy a variable from a warp lane with

higher ID relative to caller
• __syncwarp(mask)

▶ Guarantees memory ordering among threads participating in the barrier
▶ Threads within a warp that wish to communicate via memory can store to memory,

execute __syncwarp(), and then safely read values stored by other threads

Using CUDA Warp-Level Primitives

Concurrency and CUDA Streams

Overlap host and device computation with data transfers



Classic Copy-Then-Execute Model

1 cudaMemcpy(d_a, h_a, numBytes, cudaMemcpyHostToDevice);
2 kernel<<<1,N>>>(d_a);
3 cudaMemcpy(h_res, d_a, numBytes, cudaMemcpyDeviceToHost);

• Data transfer on line 1 is blocking or synchronous
▶ Host thread cannot launch the kernel until the copy is done

• Kernel launch on line 2 is asynchronous
• Data transfer on line 3 cannot begin until the kernel completes due to device-side

ordering

How to Overlap Data Transfers in CUDA C/C++
GPU Pro Tip: CUDA 7 Streams Simplify Concurrency

Overlap Host and Device Computation

1 cudaMemcpy(d_a, h_a, numBytes, cudaMemcpyHostToDevice);
2 kernel<<<1,N>>>(d_a);
3 cudaMemcpy(h_res, d_a, numBytes, cudaMemcpyDeviceToHost);

1 cudaMemcpy(d_a, h_a, numBytes, cudaMemcpyHostToDevice);
2 kernel<<<1,N>>>(d_a);
3 // Host gets work done
4 h_func(h_b);
5 cudaMemcpy(h_res, d_a, numBytes, cudaMemcpyDeviceToHost);

Utilize GPU Hardware

• Overlap kernel execution with memory copy between host and device
• Overlap execution of multiple kernels if there are enough resources
• Recent GPUs support overlapped execution

▶ Check the fields asyncEngineCount, concurrentKernels, and deviceOverlap
from the cudaDeviceProp structure

CUDA Streams

• A stream is a sequence of operations that execute on the device in the order in which
they were issued by the host
▶ Operations across streams can interleave and run concurrently

• All GPU device operations run in a stream
• The default “null” stream is used if no custom stream is specified, the default stream

is synchronizing
▶ No operation in the default stream will begin until all previously issued operations in any

stream have completed
▶ An operation in the default stream must complete before any other operation in any

stream will begin

// Both launches are on the default stream
kernel1<<< blocks, threads, bytes >>>(); // default stream
kernel2<<< blocks, threads, bytes, 0 >>>(); // stream 0

How to Overlap Data Transfers in CUDA C/C++

Using a Non-default Stream

Manipulate non-default streams from the host
cudaStream_t stream1;
cudaError_t result;
result = cudaStreamCreate(&stream1);
result = cudaStreamDestroy(stream1);

Issue a data transfer to a non-default stream
result = cudaMemcpyAsync(d_a, a, N, cudaMemcpyHostToDevice, stream1);

Specifying a stream during kernel launch is optional
kernel1<<<blocks, threads, bytes>>>(); // default/NULL stream
kernel2<<<blocks, threads, bytes, stream1>>>();

How to Overlap Data Transfers in CUDA C/C++

Non-default Streams

• Operations in a non-default stream are non-blocking with the host
• Use cudaDeviceSynchronize()

▶ Blocks host until all previously issued operations on the device have completed
• Cheaper alternatives

▶ cudaStreamSynchronize(), cudaEventSynchronize(), . . .

1 cudaStream_t stream1;
2 cudaError_t res;
3 res = cudaStreamCreate(&stream1);
4 res = cudaMemcpyAsync(d_a, a, N, cudaMemcpyHostToDevice, stream1);
5 increment<<<1,N,0,stream1>>>(d_a);
6 // Blocks the host thread
7 cudaStreamSynchronize(stream1);
8 res = cudaStreamDestroy(&stream1);

Swarnendu Biswas (IIT Kanpur) CS 610: GPU Architecture and CUDA Programming Sem 2026-27-I 94 / 152



Overlapping Kernel Execution and Data Transfers

1 for (int i = 0; i < nStreams; ++i) {
2 int offset = i * streamSize;
3 cudaMemcpyAsync(&d_a[offset], &h_a[offset], streamBytes,

cudaMemcpyHostToDevice, stream[i]);
4 kernel<<<streamSize/blockSize, blockSize, 0, stream[i]>>>(d_a, offset);
5 cudaMemcpyAsync(&h_a[offset], &d_a[offset], streamBytes,

cudaMemcpyDeviceToHost, stream[i]);
6 }

How to Overlap Data Transfers in CUDA C/C++

Overlapping Kernel Execution and Data Transfers

1 for (int i = 0; i < nStreams; ++i) {
2 int offset = i * streamSize;
3 cudaMemcpyAsync(&d_a[offset], &h_a[offset], streamBytes,

cudaMemcpyHostToDevice, stream[i]);
4 }
5 for (int i = 0; i < nStreams; ++i) {
6 int offset = i * streamSize;
7 kernel<<<streamSize/blockSize, blockSize, 0, stream[i]>>>(d_a, offset);
8 }
9 for (int i = 0; i < nStreams; ++i) {

10 int offset = i * streamSize;
11 cudaMemcpyAsync(&h_a[offset], &d_a[offset], streamBytes,

cudaMemcpyDeviceToHost, stream[i]);
12 }

Streams and Concurrency in CUDA 7+

• Prior to CUDA 7, all host threads shared the default stream
▶ The default stream implicitly synchronizes with all other streams on the device
▶ Two commands from different streams cannot run concurrently if the host thread issues

any CUDA command to the default stream between them

• CUDA 7+ provides an option to have a per-host-thread default stream
▶ Commands issued to the default stream by different host threads can run concurrently
▶ Commands in the default stream may run concurrently with commands in non-default

streams
▶ Pass the option –default-stream per-thread to nvcc to enable per-thread default

streams in CUDA 7+

GPU Pro Tip: CUDA 7 Streams Simplify Concurrency

Multi-Stream Example: Legacy Behavior

1 for (int i = 0; i < num_streams; i++) {
2 cudaStreamCreate(&streams[i]);
3 cudaMalloc(&data[i], N * sizeof(float));
4 // launch one worker kernel per stream
5 kernel<<<1, 64, 0, streams[i]>>>(data[i], N);
6 // launch a dummy kernel on the default stream
7 kernel<<<1, 1>>>(0, 0);
8 }

GPU Pro Tip: CUDA 7 Streams Simplify Concurrency

Multi-Stream Example: Per-Thread Default Stream

1 for (int i = 0; i < num_streams; i++) {
2 cudaStreamCreate(&streams[i]);
3 cudaMalloc(&data[i], N * sizeof(float));
4 // launch one worker kernel per stream
5 kernel<<<1, 64, 0, streams[i]>>>(data[i], N);
6 // launch a dummy kernel on the default stream
7 kernel<<<1, 1>>>(0, 0);
8 }

GPU Pro Tip: CUDA 7 Streams Simplify Concurrency

Achievable Concurrency

Is it possible to launch two kernels that do independent tasks concurrently?

1 // host and device initialization
2 .......
3 // launch kernel1
4 myMethod1 <<<.... >>> (params);
5 // launch kernel2
6 myMethod2 <<<.....>>> (params);

• We can launch concurrent kernels in different streams
• There must be resources available while one kernel is running to run concurrent

kernels
• The maximum concurrency is 16 kernels on Fermi, 32 on Kepler, and 128 on Turing

and Ampere

Compute Capabilities: Features and Technical Specifications

Concurrent Kernels

concurrent-kernels.cu

Achievable Concurrency

Can we run two CUDA kernels from two different applications concurrently?

• A kernel from one CUDA context (analogue of host processes for the device) cannot
execute concurrently with a kernel from another CUDA context

• The GPU may time slice to provide forward progress to each context
• Must enable Multi-Process Service (MPS) to run kernels from multiple process

simultaneously

Performance Bottlenecks in CUDA



Key Ideas for Extracting Performance

• Desired application characteristics for device execution
(i) Large data-parallel computation

(ii) Complex computation kernel to justify the H2D and D2H data movement costs
■ Keep data on the device to avoid repeated transfers

• Try and reduce resource consumption (check with -res-usage)
• Exploit SIMT, reduce thread divergence in a warp
• Strive for good locality, use tiling to exploit shared memory

▶ Improve throughput by reducing global memory traffic
▶ Copy blocks of data from global memory to shared memory and operate on them (e.g.,

matrix multiplication kernel)

• Optimize memory accesses
Global memory memory coalescing

Shared memory avoid bank conflicts

What can we say about this code?

1 __global__ void dkernel(float *vector, int vectorsize) {
2 int id = blockIdx.x * blockDim.x + threadIdx.x;
3 switch (id) {
4 case 0: vector[id] = 0; break;
5 case 1: vector[id] = vector[id] * 10; break;
6 case 2: vector[id] = vector[id - 2]; break;
7 case 3: vector[id] = vector[id + 3]; break;
8 ...
9 case 31: vector[id] = vector[id] * 9; break;

10 }
11 }

Dealing with Thread Divergence

• Thread divergence renders execution sequential because the SIMD hardware takes
multiple passes through the divergent paths

if (threadIdx.x / WARP_SIZE > 2) {}

Condition evaluating to different truth values is not bad
• Branch granularity is a whole multiple of warp size; all threads in any given warp

follow the same path

if (threadIdx.x > 2) {}

Conditions evaluating to different truth-values for threads in a warp is bad
• Creates two different control paths for threads in a block; branch granularity < warp

size; threads 0 and 1 follow different path than the rest of the threads in the first warp

Parallel Memory Architecture

• In a parallel architecture, many threads access memory
• Memory is divided into banks to achieve high bandwidth

▶ Each bank can service one address per cycle
▶ A memory can service as many simultaneous accesses as it has banks

• Multiple simultaneous accesses to a bank result in a bank conflict
− Conflicting accesses are serialized

Example of Bank Addressing

Random access

Thread 0
Thread 1
Thread 2
Thread 3

Thread 15

Bank 0
Bank 1
Bank 2
Bank 3

Bank 15

Thread 4
Thread 5
Thread 6
Thread 7
Thread 8

Bank 4
Bank 5
Bank 6
Bank 7
Bank 8

Linear access, stride=1

Thread 0
Thread 1
Thread 2
Thread 3

Thread 15

Bank 0
Bank 1
Bank 2
Bank 3

Bank 15

Thread 4
Thread 5
Thread 6
Thread 7
Thread 8

Bank 4
Bank 5
Bank 6
Bank 7
Bank 8

Example of Bank Addressing

Linear access, stride=2

Thread 0
Thread 1
Thread 2
Thread 3

Thread 15

Bank 0
Bank 1
Bank 2
Bank 3

Bank 15

Thread 4
Thread 5
Thread 6
Thread 7
Thread 8

Bank 4
Bank 5
Bank 6
Bank 7
Bank 8

Linear access, stride=8

Thread 0
Thread 1
Thread 2
Thread 3

Thread 15

Bank 0
Bank 1
Bank 2
Bank 3

Bank 15

Thread 4
Thread 5
Thread 6
Thread 7
Thread 8

Bank 4
Bank 5
Bank 6
Bank 7
Bank 8

Bank Conflicts in Shared Memory

• NVIDIA GPUs have 32 shared memory banks per SM
▶ Each 4-byte shared memory word W is mapped bank (&W/4)%32

• Shared memory is as fast as registers if there are no bank conflicts

Fast case • If all threads of a warp access different banks, there is no bank conflict
• If all threads of a warp access an identical address, there is no bank

conflict (broadcast)
Slow case • Bank conflict arises when multiple threads in the same half-warp access

the same bank
▶ Must serialize the accesses

• Cost = max # of simultaneous accesses to a single bank

Give low priority to fix low-degree bank conflicts since resolving it will increase
instructions

Using Shared Memory in CUDA C/C++
Notes About Nvidia GPU Shared Memory Banks shmem-benchmarking.cu

Memory Coalescing
• Issuing a memory instruction from a single warp can generate up to 32 data cache

accesses
• Coalescing reduces the number of memory requests by merging accesses from

multiple lanes into cache-line-sized chunks when there is spatial locality across the
warp
▶ Coalesced memory accesses imply a warp accesses adjacent data in a cache line
▶ In the best case, this results in one memory transaction

• Uncoalesced memory accesses imply a warp accesses scattered data in different
cache lines leading to memory divergence
▶ This may result in 32 different memory transactions

Copying a Matrix
j * 32

k 
* 3

2

Tile

32

32

TPB = 32 x 8
NB = j x k

Tile

32

32Tile

0

8

16

24

...

...

...

...

Each 32 × 8 thread block copies a 32 × 32 tile

1 __global__ void copy(float *odata, const float *idata) {
2 int x = blockIdx.x * TILE_DIM + threadIdx.x;
3 int y = blockIdx.y * TILE_DIM + threadIdx.y;
4 int width = gridDim.x * TILE_DIM;
5 for (int j = 0; j < TILE_DIM; j += BLOCK_ROWS)
6 odata[(y+j)*width + x] = idata[(y+j)*width + x];
7 }

An Efficient Matrix Transpose in CUDA C/C++

Matrix Transpose

1 __global__ void transposeNaive(float *odata, const float *idata) {
2 int x = blockIdx.x * TILE_DIM + threadIdx.x;
3 int y = blockIdx.y * TILE_DIM + threadIdx.y;
4 int width = gridDim.x * TILE_DIM;
5 for (int j = 0; j < TILE_DIM; j+= BLOCK_ROWS)
6 odata[x*width + (y+j)] = idata[(y+j)*width + x];
7 }

reads from idata are coalesced,
but writes to odata have a stride
of j*32

An Efficient Matrix Transpose in CUDA C/C++

Optimizing Matrix Transpose

1 __global__ void transposeCoalesced(float *odata, const float *idata) {
2 __shared__ float tile[TILE_DIM][TILE_DIM];
3 int x = blockIdx.x * TILE_DIM + threadIdx.x;
4 int y = blockIdx.y * TILE_DIM + threadIdx.y;
5 int width = gridDim.x * TILE_DIM;
6 for (int j = 0; j < TILE_DIM; j += BLOCK_ROWS)
7 tile[threadIdx.y+j][threadIdx.x] = idata[(y+j)*width + x];
8 __syncthreads();
9 x = blockIdx.y * TILE_DIM + threadIdx.x; // transpose block offset

10 y = blockIdx.x * TILE_DIM + threadIdx.y;
11 for (int j = 0; j < TILE_DIM; j += BLOCK_ROWS)
12 odata[(y+j)*width+x] = tile[threadIdx.x][threadIdx.y+j];
13 }

transpose-nvidia.cu

Matrix Multiplication Example

1 __global__ void matmulKernel(float* A, float* B, float* C) {
2 int row = blockIdx.y * blockDim.y + threadIdx.y;
3 int col = blockIdx.x * blockDim.x + threadIdx.x;
4 float tmp = 0;
5 if (row < N && col < N) {
6 // Each thread computes one element of the matrix
7 for (int k = 0; k < N; k++) {
8 tmp += A[row * N + k] * B[k * N + col];
9 }

10 }
11 C[row * N + col] = tmp;
12 }

Optimizing Global Memory Accesses

• Try to ensure that memory requests from a warp can be
coalesced
▶ Using optimizations like tiling to make use of the faster

shared memory
▶ Stride-one access across threads in a warp is good
▶ Use structure of arrays rather than array of structures

Swarnendu Biswas (IIT Kanpur) CS 610: GPU Architecture and CUDA Programming Sem 2026-27-I 115 / 152



Implementing a Reduction Kernel in CUDA

M. Harris. Optimizing Parallel Reduction in CUDA.

Reduction Kernel

1 __global__ void reduction1(const uint64_t *d_input, uint64_t *d_output) {
2 uint64_t tid = threadIdx.x;
3 uint64_t pindex = blockIdx.x * blockDim.x + threadIdx.x;
4 __shared__ uint64_t partialSum[TPB];
5 partialSum[tid] = d_input[pindex];
6 __syncthreads();
7 unsigned int t = threadIdx.x;
8 for (unsigned int stride = 1; stride < blockDim.x; stride *= 2) {
9 if (t % (2*stride) == 0)

10 partialSum[t] += partialSum[t+stride];
11 __syncthreads();
12 }
13 if (tid == 0) {
14 d_output[blockIdx.x] = partialSum[tid];
15 }
16 }

only even threads
are active

reduction.cu

Possible Optimizations on the Reduction Kernel

(i) basic implementation with modulo operator
(ii) strided access starting with each thread accessing two

adjacent locations
(iii) strided access with reversed loop index
(iv) halve the number of threads
(v) unroll the last few loop iterations and avoid

synchronization
(vi) . . .

M. Harris. Optimizing Parallel Reduction in CUDA.

Reduction Kernel

1 __global__ void reduction3(const uint64_t *d_input, uint64_t *d_output) {
2 uint64_t tid = threadIdx.x;
3 uint64_t pindex = blockIdx.x * blockDim.x + threadIdx.x;
4 __shared__ uint64_t partialSum[TPB];
5 partialSum[tid] = d_input[pindex];
6 __syncthreads();
7 for (uint64_t stride = (blockDim.x >> 1); stride >= 1; stride >>= 1) {
8 if (tid < stride) {
9 partialSum[tid] += partialSum[tid + stride];

10 }
11 __syncthreads();
12 }
13 if (tid == 0) {
14 d_output[blockIdx.x] = partialSum[tid];
15 }
16 }

Swarnendu Biswas (IIT Kanpur) CS 610: GPU Architecture and CUDA Programming Sem 2026-27-I 119 / 152



Execution of the Revised Reduction Kernel

Last five iterations
still have divergence

Avoid Divergence in the Last Few Iterations

1 for (unsigned int stride = blockDim.x/2; stride > 32; stride /= 2) {
2 if (t < stride)
3 partialSum[t] += partialSum[t+stride];
4 __syncthreads();
5 }
6 if (tid < 32) {
7 partialSum[tid] += partialSum[tid+32];
8 partialSum[tid] += partialSum[tid+16];
9 partialSum[tid] += partialSum[tid+8];

10 partialSum[tid] += partialSum[tid+4];
11 partialSum[tid] += partialSum[tid+2];
12 partialSum[tid] += partialSum[tid+1];
13 }

Efficient Data Management



Data Transfer Between CPU and GPU

• Direct Memmory Access (DMA) hardware
transfers bytes between physical memory
address spaces
▶ Uses system interconnect, typically PCIe in

today’s systems
• cudaMemcpy() copies data as one or more

DMA transfers
▶ Source and destination pointers are virtual

• Virtual memory complicates data transfer
▶ Pages in virtual address space are mapped into

and out of the physical memory
▶ The OS could accidentally page out the data that

is being accessed by a DMA unit and page in
another virtual page into the same physical
location

Enabling DMA Transfer in CUDA with Pinned Memory

Definition
Pinned pages are virtual memory pages that are specially marked so that they cannot be
paged out (also called Page-Locked Memory or Locked Pages)

CPU memory that serves as the source or destination of a DMA transfer with
cudaMemcpy() must be allocated as pinned memory
• The entire source and destination virtual address ranges are translated, and mapped

to physical pages
• The CUDA driver checks if the entire memory page is pinned (i.e., unswappable)
• If the source or destination of cudaMemcpy() in the host is not pinned, it needs to be

first copied to a pinned memory leading to extra overhead
▶ cudaMemcpy() is faster if the host memory is allocated in pinned memory because no

extra copy is needed

Swarnendu Biswas (IIT Kanpur) CS 610: GPU Architecture and CUDA Programming Sem 2026-27-I 123 / 152



Pinned Memory with CUDA

• Allocate and free pinned memory with cudaHostAlloc() and cudaFreeHost()
with the option cudaHostAllocDefault

• Pinned memory is a limited resource—over-subscription can have serious
consequences

How to Optimize Data Transfers in CUDA C/C++

cudaMemCpy() with Pageable Memory

cudaMemCpy() with Pinned Memory

Zero-Copy Memory with CUDA

• Zero copy memory is pinned memory
that is mapped into the device address
space
▶ Both host and device have

fine-grained direct access
+ Can leverage host memory when there

is insufficient device memory
+ Avoids explicit data transfers between

host and device
▶ Should be used for occasionally when

the data is read-only or the GPU
memory is really scarce

Improving GPU Memory Oversubscription Performance

cudaMemCpy() with Zero-Copy Memory

Managed Memory

• Unified Virtual Memory (UVM) provides
a single memory space accessible by
all GPUs and CPUs in the system

• Use cudaMallocManaged() to
allocate data in unified memory or use
__managed__ keyword in the global
scope
▶ Returns a pointer that can be

accessed from both host and device
code

Unified Memory in CUDA 6
An Even Easier Introduction to CUDA
Unified Memory for CUDA Beginners

Managed Memory

• UVM features have been evolving
starting from CUDA 6+
▶ 6.x: use a single pointer in both CPU

functions and GPU kernels
▶ 8.x: added 49-bit virtual addressing

and on-demand page migration

• CUDA runtime automatically migrates
data allocated in Unified Memory
between host and device, different
from UVA

Improving GPU Memory Oversubscription Performance

On-demand Paging

Advanced CUDA programming: asynchronous execution, memory models, unified memory

cudaMemcpy() with Managed Memory

cudaMemcpy() with Managed Memory

Unified Virtual Memory (UVM)

• On pre-Pascal GPUs, cudaMallocManaged()
allocated managed space on the device that
was active at the time of the call

• Pascal onward, cudaMallocManaged() does
not immediately allocate physical memory
▶ Pages and page table entries are not created

until the first access by the GPU or the CPU
• Hardware supports page faulting and

migration
▶ The GPU stalls the accessor threads when

they access absent pages
▶ The Page Migration Engine migrates pages to

the device before resuming the threads

J. Jung et al. Overlapping Host-to-Device Copy and Computation using Hidden Unified Memory. PPoPP’20.

Pre-Pascal Behavior of cudaMallocManaged()
__global__ void add(int n, float* x, float* y) {

int index = blockIdx.x * blockDim.x + threadIdx.x;
int stride = blockDim.x * gridDim.x;
for (int i = index; i < n; i += stride)

y[i] = x[i] + y[i];
}

int main(void) {
int N = (1 << 20);
float *x, *y;
cudaMallocManaged(&x, N * sizeof(float));
cudaMallocManaged(&y, N * sizeof(float));
for (int i = 0; i < N; i++)

x[i] = 1.0f; y[i] = 2.0f;
int blockSize = 256;
int numBlocks = (N + blockSize - 1) / blockSize;
add<<<numBlocks, blockSize>>>(N, x, y);
cudaDeviceSynchronize();
...

}

x and y are allocated on GPU memory,
driver sets up page table entries

Page fault on CPU, x and y
are copied to CPU memory

Lacked support for page faults, so all data
is copied to GPU before kernel launchMoves data back

to CPU memory

Unified Memory for CUDA Beginners

Pre-Pascal Behavior of cudaMallocManaged()
__global__ void add(int n, float* x, float* y) {

int index = blockIdx.x * blockDim.x + threadIdx.x;
int stride = blockDim.x * gridDim.x;
for (int i = index; i < n; i += stride)

y[i] = x[i] + y[i];
}

int main(void) {
int N = (1 << 20);
float *x, *y;
cudaMallocManaged(&x, N * sizeof(float));
cudaMallocManaged(&y, N * sizeof(float));
for (int i = 0; i < N; i++)

x[i] = 1.0f; y[i] = 2.0f;
int blockSize = 256;
int numBlocks = (N + blockSize - 1) / blockSize;
add<<<numBlocks, blockSize>>>(N, x, y);
cudaDeviceSynchronize();
...

}

x and y are allocated on GPU memory,
driver sets up page table entries

Page fault on CPU, x and y
are copied to CPU memory

Lacked support for page faults, so all data
is copied to GPU before kernel launchMoves data back

to CPU memory

− No concurrent access
− No on-demand migration to GPU
− No oversubscription

Unified Memory for CUDA Beginners

Behavior of cudaMallocManaged() for Pascal+

__global__ void add(int n, float* x, float* y) {
int index = blockIdx.x * blockDim.x + threadIdx.x;
int stride = blockDim.x * gridDim.x;
for (int i = index; i < n; i += stride)

y[i] = x[i] + y[i];
}

int main(void) {
int N = (1 << 20);
float *x, *y;
cudaMallocManaged(&x, N * sizeof(float));
cudaMallocManaged(&y, N * sizeof(float));
for (int i = 0; i < N; i++)

x[i] = 1.0f; y[i] = 2.0f;
int blockSize = 256;
int numBlocks = (N + blockSize - 1) / blockSize;
add<<<numBlocks, blockSize>>>(N, x, y);
cudaDeviceSynchronize();
...

}

Physical memory for x and y are not immediately
allocated, allocated only on first access or prefetch

Page fault on CPU, x and y are
allocated on CPU memory

Supports page faults, so no data migration
overhead before kernel launch

Page migration cost is now part
of the kernel execution time

Unified Memory for CUDA Beginners

Profiling with nvprof on Turing GPU

Swarnendu Biswas (IIT Kanpur) CS 610: GPU Architecture and CUDA Programming Sem 2026-27-I 137 / 152

https://www.cse.iitk.ac.in/users/swarnendu/courses/autumn2026-cs610/examples/cuda/add-grid.cu


Page Migration: High-Level Mechanism

• Assume a Pascal+ GPU accesses a page not present in the GPU memory
• Address translation for the faulting page generates a fault message and locks the

TLBs for the corresponding SM
▶ On some architectures, two SMs share a TLB, so both are locked
▶ Locking implies outstanding translations can proceed but new translations will be stalled

until all faults are resolved

• GPU can generate many faults concurrently for the same page
• UVM driver processes the faults, removes duplicates, updates page table mappings,

and transfers the data
• Fault handling adds significant overhead to streaming performance of UVM

Maximizing Unified Memory Performance in CUDA

Ways to Reduce Page Migration Overhead

(i) Initialize data on the device __global__ void init(int n, float *x, float *y) {
int index = threadIdx.x + blockIdx.x * blockDim.x;
int stride = blockDim.x * gridDim.x;
for (int i = index; i < n; i += stride) {

x[i] = 1.0f; y[i] = 2.0f;
}

}

Unified Memory for CUDA Beginners

Ways to Reduce Page Migration Overhead

(ii) Use warm-up iterations or take the
average of kernel multiple runs

(iiii) Prefetch data on the GPU with
cudaMemPrefetchAsync()

int device = -1;
cudaGetDevice(&device);
cudaMemPrefetchAsync(X, N*sizeof(float), device, NULL);
cudaMemPrefetchAsync(Y, N*sizeof(float), device, NULL);
cudaMemPrefetchAsync(Z, N*sizeof(float), device, NULL);

Unified Memory for CUDA Beginners

Explicit Memory Hints in UVM

• Advise runtime on expected memory access behaviors with cudaMemAdvise(ptr,
count, hint, device)

• Possible hints:
cudaMemAdviseSetReadMostly Specify read duplication

cudaMemAdviseSetPreferredLocation Suggest best location
cudaMemAdviseSetAccessedBy Suggest mapping

• Hints do not trigger data movement by themselves

CUDA Unified Memory

Explicit Memory Hints in UVM

cudaMemAdviseSetReadMostly
• Data will usually be read-only

• UVM will make a “local” copy of the data for each processor that touches it

• A write will invalidate all copies except the one written

cudaMemAdviseSetPreferredLocation
• Suggests which compute device is the best location for the data

• If possible, data mappings will be provided when other processors touch it

• If mapping is not possible, data is migrated

cudaMemAdviseSetAccessedBy
• Does not cause movement or affect location of data

• Indicated compute device receives a mapping to the data

• If the data is migrated, mapping is updated

Driver-based Prefetching

• Providing user hints can be complicated and error-prone
• CUDA driver can implement different prefetch policies: (i) random, (ii) sequential, or

(iii) locality-aware
▶ For example, “random” prefetches a random 4 KB page from the 2 MB large page

boundary along with the 4 KB faulting page
• NVIDIA implements a locality-aware tree-based neighborhood prefetcher

▶ Migrate multiples of 64 KB blocks contiguous in the virtual address space grouped in a
single transfer

▶ All pages being prefetched are local to the current faulty page and are within 2 MB large
page boundary

D. Ganguly et al. Interplay between Hardware Prefetcher and Page Eviction Policy in CPU-GPU Unified Virtual Memory. ISCA’19.
Mao Lin et al. Forest: Access-aware GPU UVM Management. ISCA’25.

Tree-based Neighborhood Prefetcher

• Allocation with cudaMallocManaged()
is logically divided into 2 MB large
pages

• The 2 MB pages are further divided into
logical 64 KB basic blocks to create a
full binary tree

• If the user-specified allocation request
is not a multiple of 2 MB, the remainder
allocation is rounded up to the next
2i ∗ 64 KB
▶ If the requests are for 4 MB and

192 KB, then two 2 MB trees and 1
256 KB trees are created

D. Ganguly et al. Interplay between Hardware Prefetcher and Page Eviction Policy in CPU-GPU Unified Virtual Memory. ISCA’19.

Note about UVM

• UVM provides a coherent view of a single virtual address
space between CPUs and GPUs with automatic data migration
via demand paging
+ Allows GPUs to access a page that resides in the CPU memory

as if it were in the GPU memory
+ Allows applications to run without worrying about the device

memory capacity

struct dataElem {
int prop1;
int prop2;
char *name;

}

• UVM aims to improve programmer productivity
+ Code is less verbose, makes it easy to work with nested data structures (think of C++

classes with dynamically allocated attributes)
• UVM kernels may have poorer performance

− Negative is the substantial cost of address translation overhead and demand paging

Handling GPU Page Faults

• When a GPU tries to access a physical memory page that is not currently resident in
device memory, a page fault is raised and GPU runtime migrates the requested page
to the GPU memory

• Page fault handling is expensive because it requires long latency communications
between the CPU and GPU over the PCIe bus
▶ The GPU runtime processes a group of page faults together to amortize overhead

H. Kim et al. Batch-Aware Unified Memory Management in GPUs for Irregular Workloads. ASPLOS’20

Memory Access Divergence

• An SIMD memory instruction cannot complete until data for all threads are available
▶ Problematic for irregular applications with little scope for coalescing
▶ Execution of one instruction requires multiple cache accesses when accesses fall on

distinct cache lines and multiple virtual-to-physical address translations when accesses
fall on distinct pages

• Negative impact from divergence can impact address translation more than cache
access
▶ Irregular memory accesses can lead from 1 to warp size (32/64) address translation

requests, most will miss in the TLB
▶ A page table walk on a TLB miss can take up to four memory accesses, for a total of

128–256 memory accesses per instruction
▶ GPUs employ physical caches which makes address translation the bottleneck

Swarnendu Biswas (IIT Kanpur) CS 610: GPU Architecture and CUDA Programming Sem 2026-27-I 147 / 152



Address Translation Request

• IOMMU is a hardware component on the CPU that
services address translation requests for
accesses to the DRAM by any accelerator (e.g.,
GPU)
▶ IOMMU supports multiple page table walkers

(e.g., 8–16) to concurrently service multiple page
table walk requests (TLB misses)

▶ IOMMU employs small page walk caches for the
first three levels of the page tables

• GPU multiprocessors (compute units) share a
private L1 TLB across SIMD units

• The GPU’s L1 TLBs are backed by a larger L2 TLB
that is shared across all the CUs in the GPU

S. Shin et al. Scheduling Page Table Walks for Irregular GPU Applications. ISCA’18.

Address Translation Request

(i) An address translation request is generated by a SIMD load/store
instruction

(ii) A hardware coalescer merges multiple requests to the same page
generated by the same SIMD instruction

(iii) The coalesced translation request looks up the GPU’s L1 TLB and
then the GPU’s shared L2

(iv) On a miss in the GPU’s L2 TLB, the request is sent to the IOMMU
(v) At the IOMMU, the request looks up the IOMMU’s TLBs

(vi) On a miss, the request queues up as a page walk request in the
IOMMU buffer

(vii) When an IOMMU’s page table walker becomes free, it selects a
pending request from the IOMMU buffer in some order

(viii) The page table walker first performs a PWC lookup and then
completes the walk of the page table, generating one to four
memory accesses

(ix) On finishing a walk, the desired translation is returned to the TLBs
and ultimately to the GPU SIMD unit that requested it

S. Shin et al. Scheduling Page Table Walks for Irregular GPU Applications. ISCA’18.

Memory Oversubscription

• UVM support allows GPUs to oversubscribe memory
• With oversubscription, a memory page is first evicted from GPU memory to system

memory, followed by transfer of requested memory from CPU to GPU

Improving GPU Memory Oversubscription Performance

Prefetching with Memory Oversubscription

• Aggressive prefetching under memory constraint can be counter-productive, may
cause displacement of heavily-accessed pages

• CUDA drivers implement LRU 4 KB page replacement policy
• Penalty of faults are greater under oversubscription

▶ Threads need to be stalled for writing back pages along with the latency to migrate new
pages

• An option is to disable prefetching on oversubscription

